NAG Fortran Library Routine Document

F07CDF (DGTTRF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07CDF (DGTTRF) computes the LU factorization of a real n by n tridiagonal matrix A.

2 Specification

```
SUBROUTINE F07CDF (N, DL, D, DU, DU2, IPIV, INFO)
INTEGER             N, IPIV(*), INFO
double precision    DL(*), D(*), DU(*), DU2(*)
```

The routine may be called by its LAPACK name ***dgttrf***.

3 Description

F07CDF (DGTTRF) uses Gaussian elimination with partial pivoting and row interchanges to factorize the matrix A as

$$A = PLU,$$

where P is a permutation matrix, L is unit lower triangular with at most one non-zero subdiagonal element in each column, and U is an upper triangular band matrix, with two superdiagonals.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

5 Parameters

1: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

2: DL($*$) – ***double precision*** array *Input/Output*

Note: the dimension of the array DL must be at least $\max(1, \mathrm{N} - 1)$.

On entry: must contain the $(n-1)$ subdiagonal elements of the matrix A.

On exit: is overwritten by the $(n-1)$ multipliers that define the matrix L of the LU factorization of A.

3: D($*$) – ***double precision*** array *Input/Output*

Note: the dimension of the array D must be at least $\max(1, \mathrm{N})$.

On entry: must contain the n diagonal elements of the matrix A.

On exit: is overwritten by the n diagonal elements of the upper triangular matrix U from the LU factorization of A.

4: DU($*$) – ***double precision*** array *Input/Output*

Note: the dimension of the array DU must be at least $\max(1, \mathrm{N}-1)$.

On entry: must contain the $(n-1)$ superdiagonal elements of the matrix A.

On exit: is overwritten by the $(n-1)$ elements of the first superdiagonal of U.

5: DU2($*$) – ***double precision*** array *Output*

Note: the dimension of the array DU2 must be at least $\max(1, \mathrm{N}-2)$.

On exit: contains the $(n-2)$ elements of the second superdiagonal of U.

6: IPIV($*$) – INTEGER array *Output*

Note: the dimension of the array IPIV must be at least $\max(1, \mathrm{N})$.

On exit: contains the n pivot indices that define the permutation matrix P. At the ith step, row i of the matrix was interchanged with row IPIV(i). IPIV(i) will always be either i or $(i+1)$, IPIV(i) $= i$ indicating that a row interchange was not performed.

7: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, u_{ii} is exactly zero. The factorization has been completed, but the factor U is exactly singular, and so division by zero will occur if it is used to solve a system of equations.

7 Accuracy

The computed factorization satisfies an equation of the form

$$A + E = PLU,$$

where

$$\|E\|_\infty = O(\epsilon)\|A\|_\infty$$

and ϵ is the ***machine precision***.

Following the use of this routine, F07CEF (DGTTRS) can be used to solve systems of equations $AX = B$ or $A^{\mathrm{T}}X = B$, and F07CGF (DGTCON) can be used to estimate the condition number of A.

8 Further Comments

The total number of floating-point operations required to factorize the matrix A is proportional to n.

The complex analogue of this routine is F07CRF (ZGTTRF).

9 Example

To factorize the tridiagonal matrix A given by

$$A = \begin{pmatrix} 3.0 & 2.1 & 0 & 0 & 0 \\ 3.4 & 2.3 & -1.0 & 0 & 0 \\ 0 & 3.6 & -5.0 & 1.9 & 0 \\ 0 & 0 & 7.0 & -0.9 & 8.0 \\ 0 & 0 & 0 & -6.0 & 7.1 \end{pmatrix}.$$

9.1 Program Text

```
*     F07CDF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=50)
*     .. Local Scalars ..
      INTEGER          I, INFO, N
*     .. Local Arrays ..
      DOUBLE PRECISION D(NMAX), DL(NMAX-1), DU(NMAX-1), DU2(NMAX-2)
      INTEGER          IPIV(NMAX)
*     .. External Subroutines ..
      EXTERNAL         DGTTRF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07CDF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read the tridiagonal matrix A from data file
*
         READ (NIN,*) (DU(I),I=1,N-1)
         READ (NIN,*) (D(I),I=1,N)
         READ (NIN,*) (DL(I),I=1,N-1)
*
*        Factorize the tridiagonal matrix A
*
         CALL DGTTRF(N,DL,D,DU,DU2,IPIV,INFO)
*
         IF (INFO.GT.0) THEN
            WRITE (NOUT,99999) 'The (', INFO, ',', INFO, ')',
     +        ' element of the factor U is zero'
         END IF
*
*        Print details of the factorization
*
         WRITE (NOUT,*) 'Details of factorization'
         WRITE (NOUT,*)
         WRITE (NOUT,*) ' Second super-diagonal of U'
         WRITE (NOUT,99998) (DU2(I),I=1,N-2)
         WRITE (NOUT,*)
         WRITE (NOUT,*) ' First super-diagonal of U'
         WRITE (NOUT,99998) (DU(I),I=1,N-1)
         WRITE (NOUT,*)
         WRITE (NOUT,*) ' Main diagonal of U'
         WRITE (NOUT,99998) (D(I),I=1,N)
         WRITE (NOUT,*)
         WRITE (NOUT,*) ' Multipliers'
         WRITE (NOUT,99998) (DL(I),I=1,N-1)
         WRITE (NOUT,*)
         WRITE (NOUT,*) ' Vector of interchanges'
         WRITE (NOUT,99997) (IPIV(I),I=1,N)
*
      ELSE
         WRITE (NOUT,*) 'NMAX too small'
```

```
      END IF
      STOP
*
99999 FORMAT (1X,A,I3,A,I3,A,A)
99998 FORMAT (1X,8F9.4)
99997 FORMAT (1X,5I9)
      END
```

9.2 Program Data

```
F07CDF Example Program Data
   5                              :Value of N
        2.1  -1.0   1.9   8.0
  3.0   2.3  -5.0  -0.9   7.1
  3.4   3.6   7.0  -6.0         :End of matrix A
```

9.3 Program Results

```
 F07CDF Example Program Results

 Details of factorization

  Second super-diagonal of U
   -1.0000   1.9000   8.0000

  First super-diagonal of U
    2.3000  -5.0000  -0.9000   7.1000

  Main diagonal of U
    3.4000   3.6000   7.0000  -6.0000  -1.0154

  Multipliers
    0.8824   0.0196   0.1401  -0.0148

  Vector of interchanges
         2        3        4        5        5
```